Jednoduchý model stability alternátoru

*: Jednoduchý model stability alternátoru

:::DESC

::: Jednoduchý model stability alternátoru.

::: \\

::: Vznik poruchy na vazebním dvojbranu a posléze její vypnutí.

::: \\

::: Přechodný děj řešen do 1  sec.

::: \\

::: Vypnutí v čase 0.45 sec., dynamická stabilita systému dosažena.

:::PARA

:::MODEL

:::DIAGRAM[]{}

:::DATA

:::PLOT[w=100, h=70, deps=1, indep=0, common=no, multiple=no, title=no]

:::PLOT[w=100, h=70, deps=2, indep=0, common=no, multiple=no, title=no]

:::PLOT[w=100, h=70, deps=3, indep=0, common=no, multiple=no, title=no]

:::PLOT[w=100, h=70, deps=4, indep=0, common=no, multiple=no, title=no]

:::PLOT[w=100, h=70, deps=1, indep=3, common=no, multiple=no, title=no]

:::ORIG

::: Karel Noháč KEE, FEL, ZČU v Plzni

*SYSTEM;

Pt0=120;   :: [MW] Mechanický výkon turbíny [MW]

Pmax1=299.69;   :: [MW] Maximální el. výkon dodávaný alternátorem

Pa1 /SIN/ B=Pmax1;

Pmax2=74.927;   :: [MW] Maximální el. výkon dodávaný alternátorem

Pa2 /SIN/ B=Pmax2;

Pmax3=227.40;   :: [MW] Maximální el. výkon dodávaný alternátorem

Pa3 /SIN/ B=Pmax3;

Tporuch=0.2;    :: [s] Doba vzniku poruchy

Tvyppor=0.45;   :: [s] Doba vypnutí poruchy

Tf=0.35;        :: [s] Časová konstanta budícího obvodu

: Pomerna hodnota budiciho proudu s respektovanim narazoveho pribuzeni

:Ib=1+1.5*(1-EXP(-(TIME-Tporuch)/Tf))*(TIME>Tporuch); :: Budící proud

Ib=1;

: Vysledny mechanicky vykon na soustroji pri zanedbani ztrat

DeltaP=-Pt0+Ib*(

       Pa1(Theta)*(TIME<Tporuch)

      +Pa2(Theta)*(TIME>=Tporuch)*(TIME<Tvyppor)

      +Pa3(Theta)*(TIME>=Tvyppor)); 

Omega0=2PI*50;   :: [rad/s] Úhlová synchronní rychlost soustrojí ::: \omega_0 = 2 \cdot \pi \cdot 50

Tm=7;            :: [s] Mechanická časová konstanta

Sng=235;         :: [MVA] Jmenovitý výkon alternátoru

: Moment setrvacnosti

Jm=Tm*Sng/(Omega0*Omega0);

Setrvacnost > C Omega = Jm;

DeltaM > J Omega = DeltaP/Omega0;

ZateznyUhel > @Int Omega,Theta;

*TR;

INIT Theta=0.39467, Omega=0;

TR 0 2;

PRINT(1001) DeltaP, Omega, Theta, Ib;

RUN;

*END;

:: DeltaP [MW] Vysledny mechanicky vykon na soustroji pri zanedbani ztrat

:: Omega [rad/s] Uhlova rychlost soustroji

:: Theta [rad] Zatezny uhel soustroji

:: Ib [rad/s] Pomerna hodnota budiciho proudu s respektovanim narazoveho pribuzeni

